

## Wechsel des Kontaktverhaltens

Ausgabe: 6 / 2003

### Problem:

In einigen Anwendungen ist es wünschenswert das Kontaktverhalten während der Analyse zu ändern. Eine Anwendung hierfür ist z.B. die Simulation abgleitender Erdplatten (Standard Kontakt), die sich anschließend wieder verhaken und damit näherungsweise mit einem „bonded“-Kontakt dargestellt werden können.

### Lösung:


Eine direkte Lösung das Kontaktverhalten während der Analyse zu ändern ist in ANSYS nicht vorgesehen. Das liegt daran, dass das Kontaktverhalten über eine Keyoption (KEYOPT(12)) eingestellt wird, die während der Analyse nicht geändert werden.

Die Idee, dies dennoch zu realisieren, basiert auf der „birth & death“ Funktionalität der Kontaktelemente: Hierfür werden **zwei** übereinander liegende Kontaktpaare mit unterschiedlichem Kontaktverhalten (z.B. KEYOPT(12)=0 und KEYOPT(12)=3) definiert. Mit dem Kommando **ekill** kann nun ein Kontaktverhalten ausgeblendet werden bzw. mit **ealive** eingeblendet werden.



### Beispiel:

Unten sehen Sie ein einfaches Beispiel zweier (gummielastisch, ebenes Problem) Platten die zunächst aufeinander abgleiten und anschließend „gebonded“ werden. Den Input hierfür finden Sie auf der nächsten Seite.

Ausgangszustand



Bonded Kontakt



Standard Kontakt

## Wechsel des Kontaktverhaltens

Ausgabe: 6 / 2003

### Input:

```
fini
/clear
/prep7

!*Geometrie
rectng,0,100,0,10
rectng,20,120,10,20

!*Elemente
et,1,182
et,2,169
et,3,171
et,4,169
et,5,171
keyopt,3,10,1
keyopt,3,5,3
keyopt,5,2,2
keyopt,5,4,1
keyopt,5,12,3

!*Material
tb,hyper,1,1,2,mooney
tbtemp,0
tbdata,,0.32,0.18,0.1

!*Vernetzung
esize,5
amesh,all

asel,s,area,,1
nsla,s,1
nsl,r,loc,y,10
esln,s
type,2
esurf
real,2
```

```
type,4
esurf
asel,s,area,,2
nsla,s,1
nsl,r,loc,y,10
esln,s
real,1
type,3
esurf
real,2
type,5
esurf
allsel

!*Randbedingung
nsl,s,loc,x
d,all,all
nsl,s,loc,x,100
nsl,r,loc,y
d,all,all
allsel



/solu
esel,s,type,,4,5
ekill,all
nsl,s,loc,y,20
nsl,r,loc,x,20,50
d,all,uy,-1
nsl,s,loc,x,120
nsl,r,loc,y,20
d,all,ux
ngeom,on
nsubst,20
allsel
solve
```

```
kbc,1
nsl,s,loc,y,20
nsl,r,loc,x,20,50
ddelete,all,all
ldread, reac, last
nsubst,1
allsel
solve

kbc,0
nsl,s,loc,x,120
d,all,ux,40
nsl,s,loc,x,20
nsl,r,loc,y,20
ddelete,all
ngeom,on
autots,on
nsubst,20
outres,all,all
allsel
solve

esel,s,type,,4,5
ealive,all
esel,s,type,,2,3
ekill,all
allsel
nsubst,1
solve

nsl,s,loc,x,120
d,all,ux,70
nsubst,40
outres,all,all
allsel
solve
```



von Mises Spannungen